
Change 1
	Function name
	f_EUTRA_IPCAN_EmergencyCall

	Reason for change
	According to the prose for test case 19.3.3, step 4 (ACK) and steps 5-19 (IMS emergency registration on 2nd PDN) can occur in parallel. However, in the case where the IPCAN recieves PDN CONNECTIVITY REQUEST before ACK has been received, it will immediately start configuring IP PTC to start IP handling on the new PDN. It will not be possible to handle ACK being received on the 1st PDN while this is ongoing.

	Summary of change
	Restructure function f_EUTRA_IPCAN_EmergencyCall() to insert a short (e.g. 0.5s) delay between step 9 (receiving PDN CONNECTIVITY REQUEST) and step 10 (starting IP) handling on new PDN. This delay is for the purpose of receiving any parallel signalling on the 1st PDN before IP handling is reconfigured.

	TTCN module
	IPCAN_EUTRA.ttcn

	MCC160 Comment
	Temporarily accepted. 
Rev1: Accepted in principle; implemented as below


Before change

 function f_EUTRA_IPCAN_EmergencyCall(EUTRA_CellId_Type p_CellId,

                                       EPS_BearerIdentity p_EpsBearerId,

                                       boolean p_ConnectionEstablished) runs on EUTRA_PTC

  { /* @sic R5s140490 change 7 MCC160 implementation: p_ConnectionEstablished sic@ */

    if (p_ConnectionEstablished) {

      f_EUTRA_508_EstablishIMSEmergencyCall_Steps9_15(p_CellId, p_EpsBearerId);  /* @sic R5s140490 change 7 MCC160 implementation sic@ */

    } else {

      f_EUTRA_508_EstablishIMSEmergencyCall_Steps2_15(p_CellId, -, p_EpsBearerId); // without triggering the IMS Emg Call !

    }

  }                   
After change

  function f_EUTRA_IPCAN_EmergencyCall(EUTRA_CellId_Type p_CellId,

                                       EPS_BearerIdentity p_EpsBearerId,

                                       boolean p_ConnectionEstablished) runs on EUTRA_PTC

  { /* @sic R5s140490 change 7 MCC160 implementation: p_ConnectionEstablished sic@ */

   var NAS_MSG_Indication_Type v_NasInd;

    var EPS_BearerIdentity v_EpsDefBearerId2 := p_EpsBearerId;

    var EPS_BearerIdentity v_EpsDedicatedBearerId := int2hex((hex2int(v_EpsDefBearerId2) + 1), 1);
    if (p_ConnectionEstablished) {

        v_NasInd := f_EUTRA_508_EstablishIMSEmergencyCall_Step9(p_CellId);

        f_Delay(0.5);

       // Steps 10 - 14

       f_EUTRA_508_EstablishIMSEmergencyCall_Steps10_14(p_CellId, v_EpsDefBearerId2, v_NasInd, true, PDN_2);

       // Step 15

       f_EUTRA_508_EstablishIMSEmergencyCall_Step15(p_CellId, v_EpsDedicatedBearerId);
          } else {

      f_EUTRA_508_EstablishIMSEmergencyCall_Steps2_15(p_CellId, -, p_EpsBearerId); // without triggering the IMS Emg Call !

    }

  }
MCC160 Implementation
	  function f_EUTRA_IPCAN_EmergencyCall(EUTRA_CellId_Type p_CellId,

                                       EPS_BearerIdentity p_EpsBearerId,

                                       boolean p_ConnectionEstablished) runs on EUTRA_PTC

  { /* @sic R5s140490 change 7 MCC160 implementation: p_ConnectionEstablished sic@ */ 
    var boolean v_StartIPfor2ndPDN := false; 
    var DRB_Identity v_DRB_Id := f_EUTRA_EpsBearerAssociatedDRB(p_EpsBearerId);
    if (v_DRB_Id != tsc_DRB2) {

      f_IP_ChangeDrbMappingEUTRA(IP, p_CellId, v_DRB_Id, PDN_2);     /* @sic R5s150648: f_IP_ChangeDrbMappingEUTRA sic@ */
    }
    if (p_ConnectionEstablished) {

      f_EUTRA_508_EstablishIMSEmergencyCall_Steps9_15(p_CellId, p_EpsBearerId, v_StartIPfor2ndPDN);  /* @sic R5s140490 change 7 MCC160 implementation sic@ */

    } else {

      f_EUTRA_508_EstablishIMSEmergencyCall_Steps2_15(p_CellId, -, p_EpsBearerId, -, v_StartIPfor2ndPDN); // without triggering the IMS Emg Call !

    }

  }

	  function f_EUTRA_IPCAN_InitialRegistration(IMS_IPCAN_CO_ORD_PORT p_Port,

                                             EUTRA_CellId_Type p_CellId ,

                                             boolean p_IsEmergencyCallConfiguration) runs on EUTRA_PTC

  { /* @sic R5-145787: new parameter p_IsXCAP sic@ */

    var boolean v_ResetIMS := false;

    f_EUTRA_SetCellPower(p_CellId, tsc_ServingCellRS_EPRE);

    // Switch on UE

    f_EUTRA_SwitchOnUEandStartIP(p_CellId, -, -, v_ResetIMS);

    f_EUTRA_IPCAN_IdleUpdated(p_Port, p_CellId); 
    if (p_IsEmergencyCallConfiguration) { /* @sic R5s150412 sic@ */

      f_EUTRA_StartIPfor2ndPDN(p_CellId, PDN_2, tsc_DRB2);  /* we use DRB2 for now; routing table can be changed later */

    }
  }

	  function f_EUTRA_IPCAN_StartProcedure(IMS_IPCAN_CO_ORD_PORT p_Port,

                                        EUTRA_CellId_Type p_CellId,

                                        IMS_TestProcedure_Type p_ProcedureType,

                                        integer p_NextAvailableEpsBearerId,

                                        boolean p_ConnectionEstablished,

                                        boolean p_IsEmergencyCall) runs on EUTRA_PTC return TriggerResult_Type

  { /* @sic R5s130681: DISCARD_UM_DATA sic@ */

    /* @sic R5s140490 change 7 MCC160 implementation: p_ConnectionEstablished sic@ */

    var TriggerResult_Type v_Result := NORMAL;

    var EPS_BearerIdentity v_EpsBearerId := int2hex(p_NextAvailableEpsBearerId, 1);

    select (p_ProcedureType) {

      case (IPCAN_InitialRegistration) {                                        // initial registration

        f_EUTRA_IPCAN_InitialRegistration(p_Port, p_CellId, p_IsEmergencyCall);

      }

      case (IPCAN_EmergencyCall_NormalService) {                                // emergency call, normal service, uses 112 as default emg number

        f_EUTRA_IPCAN_EmergencyCall(p_CellId, v_EpsBearerId, p_ConnectionEstablished);

      }

      case (IPCAN_MO_Speech_EmergencyCall) {                                    /* @sic R5s141286 change 6 sic@

                                                                                   to add an emergency call even when there is a speech call already (been put on hold);

                                                                                   NOTE: In case there is a speech call already (put on hold, e.g. 19.1.5)

                                                                                   the default EPS bearer (v_EpsBearerId) is '7'H whereas it is '6'H in normal cases

                                                                                   @sic R5s150338: no pre-configuration of DRBs sic@ */

        f_EUTRA_IPCAN_EmergencyCall(p_CellId, v_EpsBearerId, p_ConnectionEstablished);

      }

      case (IPCAN_EmergencyCall_LimitedService) {                               // emergency call, limited service

        // to be implemented

      }

      case (IPCAN_MO_SpeechCall) {                                              // MO speech call

        v_Result := f_EUTRA_IPCAN_MO_SpeechCall(p_Port, p_CellId);

      }

      case (IPCAN_MT_SpeechCall) {                                              // MT speech call

        v_Result := f_EUTRA_IPCAN_MT_SpeechCall(p_Port, p_CellId, v_EpsBearerId);

      }

      case (IPCAN_MO_VideoCall) {                                               // MO video call

        v_Result := f_EUTRA_IPCAN_MO_VideoCall(p_Port, p_CellId);

      }

      case (IPCAN_MT_VideoCall) {                                               // MT video call

        v_Result := f_EUTRA_IPCAN_MT_VideoCall(p_Port, p_CellId);

      }

      case (IPCAN_MO_AddVideo, IPCAN_MT_AddVideo) {                             // add video to existing speech call

        f_EUTRA_IPCAN_AddVideo(p_Port, p_CellId, tsc_EpsDedicatedBearerId2);

      }

      case (IPCAN_ReleaseVideo) {                                               // Release video but keep speech call

        f_EUTRA_IPCAN_ReleaseVideo(p_Port, p_CellId, tsc_EpsDedicatedBearerId2);

      }

      case (IPCAN_MO_IMS_Signalling) {                                          // MO IMS signalling

        f_EUTRA_IPCAN_MO_IMS_Signalling(p_CellId);

      }

      case (IPCAN_MT_IMS_Signalling) {                                          // MT IMS signalling

        f_EUTRA_IPCAN_MT_IMS_Signalling(p_Port, p_CellId);                      /* @sic R5s130183 change 2 sic@ */

      }

      case (IPCAN_XCAP_Signalling) {                                            // XCAP signalling

        v_EpsBearerId := tsc_EpsDefaultBearerId2ndPDN;

        f_EUTRA_IPCAN_XCAP_Signalling(p_Port, p_CellId, v_EpsBearerId);         /* @sic R5s141051 change 6: v_EpsBearerId sic@ */

      }

    }

    return v_Result;

  }

	   function f_EUTRA_IPCAN_MainLoop() runs on EUTRA_PTC

  {

    var EUTRA_CellId_Type v_CellId := tsc_IMS_EUTRA_CellId;

    var IMS_IPCAN_Coordination_MSG v_IMS_IPCAN_Coordination_MSG;

    var IMS_IPCAN_Command_Type v_Command;

    var integer v_PdnIndex;

    var boolean v_DedicatedBearersEstablished := false;  // will be set to true e.g. after successful call establishment (note: there are test cases with unsuccessful call establishment)

    var boolean v_ConnectionEstablished := false;      /* @sic R5s140490 change 7 MCC160 implementation sic@ */

    var boolean v_Done := false;

    var integer v_NextAvailableEpsBearerId := hex2int(tsc_EpsDedicatedBearerId);  // the next EPS bearer available is to be used as default for IMS Normal Emg Call

    var boolean v_IsXCAP := false; 
    var boolean v_IsEmergencyCallConfiguration := false;
    while (not v_Done) {

      alt {

        [] IMS[tsc_Index_PDN1].receive(cmr_IMS_IPCAN_Command) -> value v_IMS_IPCAN_Coordination_MSG {

          v_PdnIndex := tsc_Index_PDN1;

        }

        [] IMS[tsc_Index_PDN2].receive(cmr_IMS_IPCAN_Command) -> value v_IMS_IPCAN_Coordination_MSG {

          v_PdnIndex := tsc_Index_PDN2;

        }

      }

      v_Command := v_IMS_IPCAN_Coordination_MSG.IMS_IPCAN_Command;

      select (v_Command.Name) {

        case (IPCAN_INIT) {

          f_EUTRA_IPCAN_Init(IMS[v_PdnIndex], v_CellId);

          v_IsXCAP := (v_Command.TestConfiguration == IPCAN_XCAP);    /* @sic R5-145787: we need to know when it is an XCAP case sic@ */ 
          v_IsEmergencyCallConfiguration := (v_Command.TestConfiguration == IPCAN_EmergencyCall);
        }

        case (IPCAN_CONFIG) {

          f_EUTRA_IPCAN_ModifyConfig(IMS[v_PdnIndex], v_CellId, v_Command.CellConfiguration);

        }

        case (IPCAN_STARTPROCEDURE) {

          if (f_EUTRA_IPCAN_StartProcedure(IMS[v_PdnIndex],

                                           v_CellId,

                                           v_Command.TestProcedure,

                                           v_NextAvailableEpsBearerId,

                                           v_ConnectionEstablished ,

                                           v_IsEmergencyCallConfiguration) == NORMAL) {

            v_DedicatedBearersEstablished := true;

            select (v_Command.TestProcedure) {                       /* @sic R5s140490 change 9 MCC160 implementation: v_NextAvailableEpsBearerId to be incremented only when procedure has not been aborted sic@ */

              case (IPCAN_MO_SpeechCall, IPCAN_MT_SpeechCall, IPCAN_EmergencyCall_LimitedService) {

                v_NextAvailableEpsBearerId := v_NextAvailableEpsBearerId + 1;

              }

              case (IPCAN_MO_VideoCall, IPCAN_MO_VideoCall, IPCAN_EmergencyCall_NormalService) {

                v_NextAvailableEpsBearerId := v_NextAvailableEpsBearerId + 2;

              }

              case else {

                // no new bearer added

              }

            }

          }

          v_ConnectionEstablished := true;      /* @sic R5s140490 change 7 MCC160 implementation sic@ */

        }

        case (IPCAN_ENDPROCEDURE) {

          f_EUTRA_IPCAN_ConnectionRelease(IMS[v_PdnIndex], v_CellId, v_Command.TestProcedure, v_DedicatedBearersEstablished, v_IsXCAP);

          /* => RRC connection is released */

          v_DedicatedBearersEstablished := false;

          v_ConnectionEstablished := false;     /* @sic R5s140490 change 7 MCC160 implementation sic@ */

        }

        case (IPCAN_RELEASE) {

          f_EUTRA_IPCAN_Release(IMS[v_PdnIndex], v_CellId);

          v_Done := true;

        }

        case (IPCAN_QUERY) {  /* @sic R5s130133 additional changes: handling of queries from the IMS PTC sic@ */

          IMS[v_PdnIndex].send(cms_IPCAN_IMS_QueryResponse(f_EUTRA_IMS_IpcanInfo()));

        }

        case else {

          FatalError(__FILE__, __LINE__, "invalid command");

        }

      }

    }

  }


MCC160: Further change
	  function f_TC_19_1_5_IMS1() runs on IMS_PTC

  { /* first an IMS MO Speech Call is established and afterwards (see PTC IMS2) an IMS Emergency Call

       the configuration is done subsequently: first DRB2 and then DRBs 3 and 4 are configured */

    var IMS_InviteRequestWithSdp_Type v_InviteRequestWithSdp;  /* @sic R5w140112: IMS_InviteRequestWithSdp_Type instead of IMS_DATA_REQ */

    var INVITE_Request v_InviteRequest;

    var MessageHeader v_MessageHeader_Response183;

    var SDP_Message v_SDP_OriginalOffer;

    var SDP_Message v_SDP_HoldResumeOffer;

    var boolean v_GeolocationUsage := f_IMS_GetGeolocationUsage(); // the test case intends to use GeoLocationInformation if supported by the UE (PICS)

    f_IMS_CC_Preamble(IPCAN_EmergencyCall, IMS_REGISTERED);

    // Trigger SS to execute location related functions to allow UE to get its position

    if  (v_GeolocationUsage) {

      f_IMS_IPCAN_ModifyConfig(IPCAN, IPCAN_UpdateUELocationInformation);

    }

    // Initial conditions: Setup MO speech call

    f_IMS_CC_StartCall(IPCAN_MO_SpeechCall);

    v_InviteRequestWithSdp := f_IMS_MOCallSetup_AnnexC21_Step2_A4();

    v_InviteRequest := v_InviteRequestWithSdp.Invite;

    v_MessageHeader_Response183 := f_IMS_MOCallSetup_AnnexC21_Step3_4(v_InviteRequestWithSdp);

    f_IMS_CC_TriggerDedicatedBearerActivation(IPCAN_MO_SpeechCall);  // trigger IPCAN to continue with activation of secondary PDP context or dedicated EPS bearer

    v_SDP_OriginalOffer := f_IMS_MOCallSetup_AnnexC21_Steps5_13(v_InviteRequestWithSdp, v_MessageHeader_Response183);

    f_IMS_TestBody_Set(true);

    // User initiates hold of ongoing call

    //@siclog "Step 1" siclog@

    f_UT_CallHold(MMI);

    // Ongoing call is put on hold by UE: Steps defined in annex C.8

    //@siclog "Steps 2-5" siclog@

    v_SDP_HoldResumeOffer := f_IMS_MOCallHoldResume_AnnexC8(HOLD, v_InviteRequest, v_MessageHeader_Response183, v_SDP_OriginalOffer);

    f_IMS_PreliminaryPass(__FILE__, __LINE__, "");

    f_IMS_IMS_SendCoOrdMsg(OtherPDN);

    //@siclog "Steps 19-22" siclog@

    f_IMS_MOCallHoldResume_AnnexC8(RESUME, v_InviteRequest, v_MessageHeader_Response183, v_SDP_OriginalOffer);

    f_IMS_PreliminaryPass(__FILE__, __LINE__, "");

    // UE releases the call

    //@siclog "Steps 23-24" siclog@

    f_IMS_CallReleaseMO(v_InviteRequest);

    f_IMS_TestBody_Set(false);

    f_IMS_CC_Postamble(IPCAN_MO_Speech_EmergencyCall);  /* @sic R5s141286 change 1: IPCAN_MO_Speech_EmergencyCall sic@ */

  }


